Lezione del 10 marzo 2014

```
> restart:
> with(plots):
> newton_step := proc (x0, f, epsi, max_iter)
    # definisco le variabili locali
    local i, x, df, dx ;
    df := D(f) ;
    x  := x0 ;   # inizializza x
    for i from 1 to max_iter do
      print(i) ;
      dx := evalf(f(x)/df(x),12) ;
      x  := x - dx ;
      # controllo per terminare le iterazioni
      if ( abs(dx) < epsi ) then break ; end if ;
    end do;
    return x ;
  end:
```

```
> f := x -> x^2 - 2 ;
```

$$f := x \to x^2 - 2 \tag{1}$$

```
> newton_step( 2, f, 0.01, 10 ) ;
```

$$1$$

$$2$$

$$3$$

$$1.414215686 \tag{2}$$

```
> newton_graph := proc (x0, f, epsi, max_iter, intervallo )
    local i, x, xi, xmax, xmin, df, dx, vert, tang, A, B, C ;
    vert := [ ] ;
    tang := [ ] ;
    df   := D(f) ;
    x    := x0 ;
    xmax := x0 ;
    xmin := x0 ;
    for i from 1 to max_iter do
      dx := evalf(f(x)/df(x),20) ;
      xi := x - dx ;
      if xi > xmax then xmax := xi ; end if ;
      if xi < xmin then xmin := xi ; end if ;
      vert := [ op(vert), [ [x,0], [x,f(x)] ] ] ;
      tang := [ op(tang), [ [x,f(x)], [xi,0 ] ] ] ;
      x    := xi ;
      if abs(dx) < epsi then break ; end if ;
    end do ;
    dx := (xmax - xmin)/20 ;
    A  := plot(vert, color=blue):
    B  := plot(tang, color=red):
    C  := plot(f,intervallo[1]..intervallo[2], color=green,
  thickness=2):
    display({A,B,C});
  end:
```

```
> f := (x) -> x^2 - 2 ;
```

$$f := x \to x^2 - 2 \tag{3}$$

```
> newton_graph( 2, f, 0.01, 100, [1, 3] ) ;
```




```
> f := (x) -> signum(x)*sqrt(abs(x)) ;
```

$$f := x \rightarrow \mathrm{signum}(x)\sqrt{|x|} \qquad \textbf{(4)}$$

```
> plot( f(x),x=-1..1) ;
```




```
> D(f) ;
```

$$x \rightarrow \mathrm{signum}(1, x)\sqrt{|x|} + \frac{1}{2}\,\frac{\mathrm{signum}(x)\,\mathrm{abs}(1, x)}{\sqrt{|x|}} \qquad \textbf{(5)}$$

```
> newton_graph( 2, f, 0.01, 10, [-1, 3] ) ;
```




```
> f := x -> (x-1)^2*exp(x) ;
```

$$f := x \to (x-1)^2 \, e^x \qquad \textbf{(6)}$$

```
> plot( f(x),x=0..2) ;
```




```
> newton_graph( 2, f, 0.01, 10, [0, 2] ) ;
```




```
> f := x -> arctan(x) ;
```

$$f := x \to \arctan(x) \tag{7}$$

```
> plot( f(x),x=-2..2) ;
```




```
> newton_graph( 1.5, f, 0.01, 3, [-3, 3] ) ;
```




```
> newton_graph( 1.2, f, 0.01, 3, [-3, 3] ) ;
```




```
> EQ := -x = x-arctan(x)/D(arctan)(x) ;
  fsolve( EQ, {x=1.5} ) ;
```

$$EQ := -x = x - \arctan(x)\left(x^2 + 1\right)$$

$$\{x = 1.391745200\} \quad \textbf{(8)}$$

```
> newton_graph( 1.391745200, f, 0.01, 100, [-3, 3] ) ;
```


1
0.5
−3
−2
−1
0
1
2
3
−0.5
−1